

Висша математика

втора част

Лабораторно упражнение

Екстремум на функция на две променливи

21.04.2020

1. Достатъчно условие за екстремум

Теорема 1.1 Нека функция $f(P)$ е два пъти диференцируема в околност на стационараната точка P_0 .

Ако $\Delta(P_0) > 0$ и $f_{xx}(P_0) > 0 \Rightarrow f_{\min}(P_0)$.

Ако $\Delta(P_0) > 0$ и $f_{xx}(P_0) < 0 \Rightarrow f_{\max}(P_0)$.

Ако $\Delta(P_0) < 0 \Rightarrow f_{\text{sad}}(P_0)$.

При $\Delta(P_0) = 0$ не може да се направи заключение дали функцията $f(P)$ има екстремум без да се използват производни от по-висок ред.

2. Дефиниция на Хесиан

Нека $f(P)$ е два пъти диференцируема функция в някаква област Ω . Хесианът

$$H : f \rightarrow \begin{pmatrix} f_{xx} & f_{xy} \\ f_{yx} & f_{yy} \end{pmatrix} \Leftrightarrow Hf = \begin{pmatrix} f_{xx} & f_{xy} \\ f_{yx} & f_{yy} \end{pmatrix}$$

дефинираме със средства на езика Mathematica по следния начин

$$H[f_]:= \{\{D[f[x,y],x,x],D[f[x,y],x,y]\}, \\ \{D[f[x,y],y,x],D[f[x,y],y,y]\}\};$$

Детерминантата на Хесе

$$\Delta(x,y) = \det Hf = \begin{vmatrix} f_{xx} & f_{xy} \\ f_{yx} & f_{yy} \end{vmatrix}$$

пресмятаме чрез

$$\Delta[x_ ,y_]:= Evaluate[Det[H[f]]].$$

Обърнете внимание, че когато се изпълнява $H[f]$ функцията f може все още да няма дефиниция.

Пример 2.1 Тук $H[f]$ и $\det Hf$ са разпечатани без f да бъде дефинирана

$$Hf = \begin{pmatrix} f^{(2,0)}[x,y] & f^{(1,1)}[x,y] \\ f^{(1,1)}[x,y] & f^{(0,2)}[x,y] \end{pmatrix}$$

$$\Delta(P) = f^{(2,0)}[x,y]f^{(0,2)}[x,y] - f^{(1,1)}[x,y]^2$$

3. Решаване на уравнения в Mathematica

Чрез операторът *Solve* могат да се решават както уравнения, така и неравенства. Ще разгледаме някои примери за употреба на оператора *Solve*.

Пример 3.1 Неизвестното е скалар.

```
Solve[x^4 - 3x^3 + 3x^2 - 3x + 2 == 0, x]  
{{x -> -i}, {x -> i}, {x -> 1}, {x -> 2}}
```

Пример 3.2 Уравнението е представено във векторна форма. Решението представлява четири наредени двойки.

```
v = {x^3 - 3x + 2y, y^2};  
Solve[v == {0, 1}, {x, y}]  
{{x -> -2, y -> 1}, {x -> -1, y -> -1},  
{x -> 1, y -> 1}, {x -> 2, y -> -1}}
```

Пример 3.3 Уравнението е представено в матрична форма. Решението представлява четири наредени четворки.

```
X = {{x^2, 1}, {y^2, t}};  
A = {{2, s}, {-1, 2}};  
B = IdentityMatrix[2];  
Solve[A.X == B, {x, y, t, s}]  
{{x -> -1, y -> -1/sqrt(2), t -> 1, s -> -2},  
{x -> 1, y -> -1/sqrt(2), t -> 1, s -> -2},  
{x -> -1, y -> 1/sqrt(2), t -> 1, s -> -2},  
{x -> 1, y -> 1/sqrt(2), t -> 1, s -> -2}}
```

Извличане на данни от оператор *Solve*.

Пример 3.4 Чрез оператора *Dimensions* определяме колко на брой решения имаме и каква е тяхната размерност. В този пример имаме 4 на брой решения, всяко от които е вектор с 4 координати.

```
X = {{x^2, 1}, {y^2, t}};  
A = {{2, s}, {-1, 2}};  
B = IdentityMatrix[2];  
Solutions = Solve[A.X == B, {x, y, t, s}];  
n = Dimensions[Solutions][[1]];  
Print["Брой на решенията n=", n];  
i = 3;  
Solution = {x, y, t, s} /. Solutions[[i]];  
Print["Всички решения", Solutions];  
Print["Всички решения в матрична форма=", MatrixForm[Solutions]];  
Print["Решение No", i, "=", Solution];  
Print["Размерности на списъка от решения", Dimensions[Solutions]];
```

Брой на решенията n=4

Всички решения $\left\{ \left\{ x \rightarrow -1, y \rightarrow -\frac{1}{\sqrt{2}}, t \rightarrow 1, s \rightarrow -2 \right\}, \right.$

$\left\{ x \rightarrow 1, y \rightarrow -\frac{1}{\sqrt{2}}, t \rightarrow 1, s \rightarrow -2 \right\}, \left\{ x \rightarrow -1, y \rightarrow \frac{1}{\sqrt{2}}, t \rightarrow 1, s \rightarrow -2 \right\}, \left\{ x \rightarrow 1, y \rightarrow \frac{1}{\sqrt{2}}, t \rightarrow 1, s \rightarrow -2 \right\} \right\}$

Всички решения в матрична форма=
$$\begin{pmatrix} x \rightarrow -1 & y \rightarrow -\frac{1}{\sqrt{2}} & t \rightarrow 1 & s \rightarrow -2 \\ x \rightarrow 1 & y \rightarrow -\frac{1}{\sqrt{2}} & t \rightarrow 1 & s \rightarrow -2 \\ x \rightarrow -1 & y \rightarrow \frac{1}{\sqrt{2}} & t \rightarrow 1 & s \rightarrow -2 \\ x \rightarrow 1 & y \rightarrow \frac{1}{\sqrt{2}} & t \rightarrow 1 & s \rightarrow -2 \end{pmatrix}$$

Решение No3= $\left\{ -1, \frac{1}{\sqrt{2}}, 1, -2 \right\}$

Размерности на списъка от решения {4, 4}

4. Конкретни програмни реализации

Пример 4.1 Дадена е функцията $f(x, y) = 27xy(1 - x - y)$. Да се изследва функцията за екстремуми и седлови точки.

Екстремум на функция на две променливи

Функции които имат безбройно много стационарни точки не могат да бъдат анализирани с тази програма

Програмата не проверява дали дадена стационарна точка принадлежи на D_f

```
Clear[f, Δ, H, x, y];
(* Дефиниция на функцията f(p) *)
f[x_, y_] := 27 (1 - x - y) x y;
(* За градиента използваме
вградена функция Grad[f[x,y],{x,y}] *)
Print["∇f=", Simplify[Grad[f[x, y], {x, y}]]];
(* Дефиниция на Хесиана *)
H[f_] := {{D[f[x, y], x, x], D[f[x, y], x, y]},
           {D[f[x, y], y, x], D[f[x, y], y, y]}};
(* Операторът Evaluate показва,
че трябва да се премине към променливи x и y. *)
Hess[x_, y_] := Evaluate[H[f]];
Δ[x_, y_] := Evaluate[Det[H[f]]];
(* Определяне на стационарните точки *)
Solutions = Solve[Grad[f[x, y], {x, y}] == {0, 0}, {x, y}];
(* n е броя на решенията. *)
n = Dimensions[Solutions][[1]];
(* Дефинираме множество от точки *)
Points = Table[{0, 0}, {i, 1, n}];
Print["Стационарни точки"];
(* Оператор /. служи за подмяна на неизвестните
с техните стойности. Чете се ReplaceAll *)
For[i = 1, i ≤ n, i++,
  Points[[i]] = {x, y} /. Solutions[[i]];
  {xi, yi} = Points[[i]];
  (* Изключване на точките с комплексни координати *)
  If[Abs[Im[xi]] + Abs[Im[yi]] == 0,
    Print["P", i, "=", Points[[i]]]
  ];
];
Print["Hf=", MatrixForm[Simplify[H[f]]]];
```

```

Print[" $\Delta(P)$ =", Simplify[ $\Delta[x, y]$ ]];
(* Изследване на поведението
на функцията в стационарните точки *)
For[i = 1, i ≤ n, i++,
  (* Избор на конкретна точка Pi *)
  {xi, yi} = Points[[i]];
  (* Проверка дали Pi(xi,yi) е точка с реални координати *)
  If[Abs[Im[xi]] + Abs[Im[yi]] == 0,
    (* Проверка дали Pi(xi,yi) е точка,
    в която f(P) има минимум *)
    If[And[ $\Delta[xi, yi] > 0$ , Hess[xi, yi][[1, 1]] > 0],
      Print["fmin[P", i, "]=", f[xi, yi]]];
    (* Проверка дали Pi(xi,yi) е точка,
    в която f(P) има максимум *)
    If[And[ $\Delta[xi, yi] > 0$ , Hess[xi, yi][[1, 1]] < 0],
      Print["fmax[P", i, "]=", f[xi, yi]]];
    (* Проверка дали точката Pi(xi,yi) е седлова *)
    If[ $\Delta[xi, yi] < 0$ , Print["fsad[P", i, "]=", f[xi, yi]]];
    (* Ако  $\Delta(Pi)=0$  не можем да направим заключение *)
    If[ $\Delta[xi, yi] == 0$ ,
      Print["f?[P", i, "]=", f[xi, yi]]]]
  ];
(* Блок за начертаване на графиката на z=f(P) *)
(* Графика на функцията f(P) в кръг с радиус равен
на 2 и център началото на координатната система *)
Plot3D[f[x, y], {x, -2, 2}, {y, -Sqrt[4 - x^2], Sqrt[4 - x^2]},
  AxesLabel → {x, y, "f(P)"}, BoxRatios → {1, 1, 0.7},
  AxesStyle → Directive[Red, Italic, 18],
  PlotRange → {{-2, 2}, {-2, 2}, {-3, 3}},
  BoundaryStyle → Directive[Red, Thick]]

```

$$\nabla f = \{ -27 y (-1 + 2 x + y) , -27 x (-1 + x + 2 y) \}$$

Стационарни точки

$$P1 = \{ 0, 1 \}$$

$$P2 = \left\{ \frac{1}{3}, \frac{1}{3} \right\}$$

$$P3 = \{ 1, 0 \}$$

$$P4 = \{ 0, 0 \}$$

$$Hf = \begin{pmatrix} -54 y & -27 (-1 + 2 x + 2 y) \\ -27 (-1 + 2 x + 2 y) & -54 x \end{pmatrix}$$


$$\Delta(P) = -729 (4 x^2 + (1 - 2 y)^2 + 4 x (-1 + y))$$

$$f_{\text{sad}}[P1] = 0$$

$$f_{\text{max}}[P2] = 1$$


$$f_{\text{sad}}[P3] = 0$$

$$f_{\text{sad}}[P4] = 0$$



(* Графика на функцията $f(P)$ в равнобедрен правоъгълен триъгълник с катети равни на единица и връх при правия ъгъл в началото на координатната система *)

```
Plot3D[f[x, y], {x, 0, 1}, {y, 0, 1 - x},  
AxesLabel → {x, y, "f(P)"}, BoxRatios → {1, 1, 1},  
AxesStyle → Directive[Red, Italic, 18],  
PlotRange → {{0, 1}, {0, 1}},  
BoundaryStyle → Directive[Red, Thick]]
```



5. Задачи за изпълнение

5.1 Дадена е функцията

$$f(x, y) = 2x + xy - y.$$

Да се изследва функцията за екстремуми и седлови точки.

5.2 Дадена е функцията

$$f(x, y) = \frac{x^2 y}{2} + xy + x^2 + 3x - 1.$$

Да се изследва функцията за екстремуми и седлови точки.

Изпращайте решенията в pdf формат. В случай, че домашната работа се нуждае от поправка, pdf файлът ще ви бъде върнат със съответните корекции.